	First Steps in Prolog 





In this tutorial we will go through few examples to show basic ideas behind Prolog programming. But first, just few notes about using Prolog systems.
A Prolog program is a set of procedures (the order is indifferent), each procedure consists of one or more clauses (the order of clauses is important). There are two types of clauses: facts and rules, you will understand the difference soon. The program must be written in advance in some text processor like Emacs. Only some Prolog systems contain IDE (Integrated Development Environment). When the Prolog system is started, you will see a prompt, usually in the following form:
?-

The system is waiting for your commands/questions. First, the program should be loaded to the Prolog database. You can load a program into the database by using the consult command in the following way (do not forget the dot at the end!):
?-consult('name_of_the_file_with_the_program').

The consult command adds the clauses and facts from the specified text file to the clauses and facts already stored in the Prolog database. Thus you can load more programs into the database but be careful whether the programs do not use the procedures with the same name. Otherwise, because of cumulating of clauses, these procedures could behave incorrectly.
You can also use the reconsult command to load the program
?-reconsult('name_of_file_with_the_program').

that behaves like the consult command (it adds procedures into the database) but if there is a procedure in the database with the same name as any procedure in the reconsulted file, then the former procedure is replaced by the new definition. You will usually use the reconsult command to update the Prolog database after change of the program.
It is possible to find out the actual contents of the Prolog database by using the listing command:
?-listing.

Your Prolog program is started by calling some procedure of the program in the following way:
?-procedure_name(parameters).

Note, that you have already called a procedure when you consulted or reconsulted the file. "Calling a procedure" is called "asking a question" in Prolog.
And how do you stop the Prolog system? Simply use the command halt (again with the dot at the end):
?-halt.

S

	Genealogy Database 





A genealogy database seems to be a good introductory to Prolog. In this lesson, we present a simple Prolog program that captures basic family relations. The program demonstrates features of Prolog like using facts, rules, variables or recursion.


Tip: Use Copy&Paste function of the browser to move the code directly into the PROLOG environment. 



First, we express the property of being a man or woman by following PROLOG facts:
man(adam). 

man(peter). 

man(paul). 

  

woman(marry). 

woman(eve). 

Then, we can add relation "parent" which associates parent and child.
parent(adam,peter). % means adam is parent of peter 

parent(eve,peter). 

parent(adam,paul). 

parent(marry,paul). 

Till now, we added only facts to our program but the real power of Prolog is in rules. While facts state the relation explicitely, rules define the relation in a more general way. Each rule has its head - name of the defined relation, and its body - a real definition of the relation. The following rules define the relations being a father and being a mother using previously defined relations of being a man or woman and being a parent.
father(F,C):-man(F),parent(F,C). 

mother(M,C):-woman(M),parent(M,C). 

Note that we use variables (start with capital letter) to express the feature that every man which is a parent of any child is also her or his father. If some parameter of the relation is not important we can use anonymous variable (denoted _ ) like in these definitions:
is_father(F):-father(F,_). 

is_mother(M):-mother(M,_). 

Before proceeding further one should know how to run the Prolog programs. You run the program by asking questions like this one:
?-father(X,paul). 

which expresses: who is father of paul? The answer is X=adam, naturally.
Now extend your facts database and try to define other family relations like being a son, aunt or grandparent. Also, try to ask Prolog system various questions and see what happens. You can compare your program with following rules:
son(S,P):-man(S),parent(P,S). 

daughter(D,P):-woman(D),parent(P,D). 

  

siblings(A,B):-parent(P,A),parent(P,B),A\=B. 

% siblings have at least one common parent 

% the test A\=B preserves that siblings are different persons 

  

full_siblings(A,B):- 

parent(F,A),parent(F,B),
parent(M,A),parent(M,B),
A\=B, F\=M.
% full siblings have common parents (both)
% the test F\=M preserves that full siblings have two different parents (father and mother, naturally)
  

full_siblings2(A,B):- 

father(F,A),father(F,B),
mother(M,A),mother(M,B),
A\=B.
% another solution to "full siblings problem" that uses relations father and mother 

  

uncle(U,N):-man(U),siblings(U,P),parent(P,N).
aunt(A,N):-woman(A),siblings(A,P),parent(P,N).
  

grand_parent(G,N):-parent(G,X),parent(X,N). 

Till now, we use only one rule to express the newly defined relation but we can also define the relation using two and more rules. If we want to express that being a human means being a man or being a woman, we can do it by these two rules:
human(H):-man(H). 

human(H):-woman(H). 

The body of rule can also use the relation that is just being defined. This features is called recursion and the following rules show its typical usage:
descendent(D,A):-parent(A,D). 

descendent(D,A):-parent(P,D),descendent(P,A). 

One can use the feature of PROLOG of non-determing the input and output variables and easily define the relation ancestor:
ancestor(A,D):-descendent(D,A). 

	Representing Data Structures 





This lesson covers data structures in Prolog. The basic data structure in Prolog is term which is expressed in form name(arguments...). If the number of arguments is zero then we speak about atom. A special type of atom is number. 



Basics (Dates example)
In this section, we introduce a data structure date(Year,Month,Day) that represents date. First, we need a "constructor" of data structure date that makes the data structure from year, month and day:
make_date(Y,M,D,date(Y,M,D)).
Second, we define functions to access components of data structure in a following way:
get_year(date(Y,_,_),Y).

get_month(date(_,M,_),M).

get_day(date(_,_,D),D).
get_xxx can be used to test or generate correspondent component of data structure, but it can not be used to set the value of the component. So, we have to define set_xxx to set values of components of data structure date.
set_year(Y,date(_,M,D),date(Y,M,D)).

set_month(M,date(Y,_,D),date(Y,M,D)).

set_day(D,date(Y,M,_),date(Y,M,D)).
Now, it is easy to find the "same" day in next or previous year respectively using get and set functions.
next_year(Today,NextYear):-

        get_year(Today,Y), NY is Y+1, set_year(NY,Today,NextYear).

prev_year(Today,PrevYear):-

        get_year(Today,Y), PY is Y-1, set_year(PY,Today,PrevYear).
Note, that following definition of prev_year using next_year is not correct. Do you know why?
prev_year(Today,PrevYear):-next_year(PrevYear,Today). % incorrect
Finding next year is relatively easy but what about finding next day, i.e., tomorrow? Study following program to find where the possible problems are hidden. The definition of test correct_day follows the next section that covers working with lists.
tomorrow(Today,Tomorrow):-

        get_day(Today,D), ND is D+1, set_day(ND,Today,Tomorrow),

        correct_date(Tomorrow).

        % day inside month (i.e., not last day of month)

tomorrow(Today,Tomorrow):-

        get_month(Today,M), NM is M+1,

        set_month(NM,Today,Tmp), set_day(1,Tmp,Tomorrow),

        correct_date(Tomorrow).

        % last day of month

tomorrow(Today,Tomorrow):-

        get_year(Today,Y), NY is Y+1, make_date(NY,1,1,Tomorrow).

        % last day of year
Note that it is also possible and probably more reasonable to encapsulate the test correct_date to definitions of make_date and set_xxx.


Lists
List is a widely used data structure which is build in Prolog. It is still a term, e.g., [1,2,3] is equivalent to '.'(1,'.'(2,'.'(3,nil))). The following functions enable access to list elements.
head(H,[H|_]).

tail(T,[_|T]). % T is list
It is easy to access the first element of list as it corresponds to the head. However, finding the last element is a time consuming process as one has to go through the whole list to find it. Note that following "procedures" can be used to find the first/last element of list as well as to test whether given element is first/last element of list. It could even be used to generate list with given first/last element.
first(F,[F|_]). % the same as head

last(L,[L]).

last(L,[H|T]):-last(L,T).
The similar conclusion also holds for finding prefix and suffix respectively. Again, the same procedure can be used to test or generate prefix/suffix respectively as well as to generate list with given prefix/suffix. Try it.
prefix([],_).

prefix([H|T1],[H|T2]):-prefix(T1,T2).

suffix(S,S).

suffix([H|T],L):-suffix(T,L).
Testing membership is an important method for working with lists. Prolog definiton of member can test membership relation as well as generate successive members of list. A similar function, nth_member, can also be used to test or to generate n-th member of list. However, it can not be used to count a sequence number of given element (define the function that counts a sequence number of given element as your homework).
member(X,[X|_]).

member(X,[_|T]):-member(X,T).

nth_member(1,[M|_],M).

nth_member(N,[_|T],M):-N>1, N1 is N-1, nth_member(N1,T,M).
Another popular function on lists is append which appends a list to another list. It can be also used to disjoint list (see following definition of prefix and suffix).
append([],L,L).

append([H|T],L,[H|LT]):-append(T,L,LT).
Now, prefix and suffix relations can be easily redefined using append:
prefix(P,L):-append(P,_,L).

suffix(S,L):-append(_,S,L).
append can be successfuly used in many other operations with lists including testing (or generating) sublist. The following rule exploits again the declarative character of Prolog.
sublist(S,L):-append(_,S,P),append(P,_,L).
There are (at least) two other ways how to define sublist, e.g., using prefix and suffix relations. All these definitions are equivalent. However, the procedure sublist3 is probably the closest to traditional (non-declarative) programming style as it uses technique known as "floating window".
sublist2(S,L):-prefix(P,L),suffix(S,P).

    

   sublist3(S,L):-prefix(S,L).

sublist3(S,[_|T]):-sublist3(S,T).
  



Back to Dates example
Let us return to our example of data structure date. Now, we can define the test correct_date using lists.
First, we add two facts to Prolog database with distribution of days:
year(regular,[31,28,31,30,31,30,31,31,30,31,30,31]).

year(leap,[31,29,31,30,31,30,31,31,30,31,30,31]).
Then, we prepare the test of leap years (simplified version):
year_type(Y,leap):-

        Z is Y mod 4, Z=0. % every fourth year is leap (simplied)

year_type(Y,regular):-

        Z is Y mod 4, Z\=0.
Finally, it is possible to test correctness of date:
correct_date(date(Y,M,D)):-

        correct_month(M),

        correct_day(Y,M,D).

correct_month(M):- M>0, M<13.

correct_day(Y,M,D):-

        year_type(Y,Type),

        test_day_of_year(Type,M,D).

test_day_of_year(Type,M,D):-

        year(Type,Days),

        nth_member(M,Days,Max),

        D>0, D=<Max.
	Prolog Data Structures 





Term is a basic data structure in Prolog, i.e., everything including program and data is expressed in form of term. There are four basic types of terms in Prolog: variables, compound terms, atoms and numbers. The following picture shows the correlation between them as well as examples of corresponding terms:
term

       |-- var (X,Y)

 |

  -- nonvar (a,1,f(a),f(X))

       |-- compound (f(a),f(X))

       |

        -- atomic  (a,1)

             |-- atom (a)

             |

              -- number (1)



It is possible to use predicates var, nonvar, compound, atomic, atom and number to test the type of given term (see copy_term bellow). Prolog also provides build-in predicates to access structure of the nonvar terms as well to construct terms. 

arg(N,Term,Arg) 

- gets N th argument of term (arg(2,f(a,b,c),X) -> X=b)
- sets N th argument of term (arg(2,f(X,Y,Z),b) -> f(X,b,Z)) 

functor(Term,Functor,NumberOfArgs) 

- gets functor name and number of arguments from the term (functor(f(a,b),F,N) -> F=f,N=2) 

- constructs term with given functor and number of free arguments (functor(F,f,2) -> F=f(_,_)) 

=.. 

- decomposes the structure of term into list (f(a,b)=..X -> X=[f,a,b]) 

- constructs term from given list (T=..[f,a,X] -> T=f(a,X)) 

name(Text,List) 

- converts name into list of codes (name(abc,Y) -> Y=[97,98,99]) 

- constructs name from list of codes (name(X,[97,98,99]) -> X=abc) 

If one needs to copy a term (copy has the same structure as the original term but introduces new variables), it is possible to use predicate copy_term/2 which is build-in in most Prolog systems. However, it is straightforward to write a code of copy_term in Prolog using above mentioned predicates.
copy_term(A,B):-cp(A,[],B,_).

cp(A,Vars,A,Vars):-

   atomic(A).

cp(V,Vars,NV,NVars):-

   var(V),register_var(V,Vars,NV,NVars).

cp(Term,Vars,NTerm,NVars):-

   compound(Term),

   Term=..[F|Args],    % decompose term

   cp_args(Args,Vars,NArgs,NVars),

   NTerm=..[F|NArgs].  % construct copy term

cp_args([H|T],Vars,[NH|NT],NVars):-

   cp(H,Vars,NH,SVars),

   cp_args(T,SVars,NT,NVars).

cp_args([],Vars,[],Vars).

During copying one has to remeber copies of variables which can be used further during copying. Therefore the register of variable copies is maintained.
register_var(V,[X/H|T],N,[X/H|NT]):-

   V\==X,         % different variables

   register_var(V,T,N,NT).

register_var(V,[X/H|T],H,[X/H|T]):-

   V==X.          % same variables

register_var(V,[],N,[V/N]).

Here is an example that clarifies the notion of term copy:
f(X,g(X)) is copy of f(Y,g(Y)) but not of f(U,g(V)).

Unification
Unification is an engine of Prolog. It tries to fing most general substitution of variables in two terms such that after applying this substitution to both terms, the terms became the same. To unify terms A and B, one can easily invoke build-in unification A=B. Try to unify different terms to see what the notion "unification" really means. Again, it is straightworfard to write Prolog code of unification (we use '=' to test equality of two atomic terms or to unify variable with term only).
unify(A,B):-

   atomic(A),atomic(B),A=B.

unify(A,B):-

   var(A),A=B.            % without occurs check

unify(A,B):-

   nonvar(A),var(B),A=B.  % without occurs check

unify(A,B):-

   compound(A),compound(B),

   A=..[F|ArgsA],B=..[F|ArgsB],

   unify_args(ArgsA,ArgsB).

unify_args([A|TA],[B|TB]):-

   unify(A,B),

   unify_args(TA,TB).

unify_args([],[]).

Did you find out what is it "occurs check"? OK, try to unify the following terms X and f(X). What happens? Most Prolog systems will fill in the whole memory as they will try to construct infinite term f(f(f(...))) which should be the result of the unification. Such Prolog systems does not incorporate occurs check because of its time consuming nature. So, occurs check tests the occurence of the variable X in the term T (which is not a variable) during unification of X and T.

Operators
Writing terms in the form functor(arg1,arg2,...) is not often appropriate from the human point of view. Just compare the following two transcriptions of the same Prolog clause:
p(X,Z):-q(X,Y),r(Y,Z),s(Z).

   ':-'(p(X,Z),(','(q(X,Y),','(r(Y,Z),s(Z))))).

Which one do you prefer?
To simplify the entry of terms, Prolog introduces operators which enable "syntactic sugar", i.e., more natural way of writing terms. Operators are used with unary and binary terms only. They enable to set the location of functor (prefix, infix, postfix), the associative feature and, finally, the priority among operators.
op(Priority,Appearence,Name)

       |        |

       |         -- xfy, yfx, xfx, fx, fy, xf, yf

        -- the higher number the priority has, the lower priority

Instead of explaining the meaning of above definition, look at the following example.
op(400,yfx,'*').  % a*b*c means ((a*b)*c)

op(500,yfx,'+').

op(500,yfx,'-').  % be careful a-b-c means ((a-b)-c)

op(700,xfx,'=')   % it is not possible to write a=b=c

op(900,fy,not).   % one can write not not a and it means not(not(a))

not 1=2+3+4*5 is equivalent to:

     not(1=((2+3)+(4*5)))

     not('='(1,'+'('+'(2,3),'*'(4,5)))).

Note, that the numbers indicating priority can be different in miscellaneous implementations of Prolog (the numbers in above example are taken from Open Prolog for Macintosh).
Important!
The definition of operator is not a new program for operator but the "call" of the goal op. Thus, if you want to define an operator in the program, you write :-op(400,yfx,'*').
	List Processing 





List is a data structure directly supported in Prolog via operations for accessing head and tail of the list. However, list is still a traditional Prolog term, if one uses the obvious dot notation. We described manipulation with lists in the section "Representing Data Structures" so we just look at differential lists here and we proceed to chapters presenting areas of using lists.



Differential lists
The main problem of Prolog implementation of lists is their one-way nature. It means that if one wants to access the n-th element, he/she has also to access all the previous elements in the list. In particular, if one wants to add an element to the end of the list, it is necessary to go through all elements in the list as the following programm shows (compare it with the implementation of append):
add2end(X,[H|T],[H|NewT]):-add2end(X,T,NewT).

add2end(X,[],[X]).

But there is a technique, called differential lists, that enable appending lists or adding element to the end of list in one step. Nevertheless, it should be noted that this technique does not remove the disadvantage of visiting all previous elements in the list if the n-th element is accessed.
A differential list consist of two parts A-B and represents the list that is obtained from A by removing the tail B, e.g., [1,2,3,4]-[3,4] represents the list [1,2]. Ofcourse, if both lists A and B are ground, than there is no advantage of using differential lists, but if this technique is combined with the advantages of free variables and unification we can get impressive results. Namely, the list [1,2] can be represented by the differential list [1,2|X]-X.
If we standardize differential lists in the later way, we can write "one-step" procedures for appending lists and adding element to the end of the list:
append(A-B,B-D,A-D).

add2end(X,A-B,A-NewB):-B=[X|NewB].

Note, that the other list operations, e.g., member, have also to be rewritten to work with differential lists.

	Arithmetic Expressions 





In this lecture, we will work with arithmetic expressions in a symbolic manner which is natural for Prolog. First we write a program for evaluating arithmetic expressions and then we develop a simple compiler which translates the arithmetic expression into a linear code of stack machine.

Evaluating
We can easily evaluate the arithmetic expression using Prolog build-in evaluator.
naive_eval(Expr,Res):-Res is Expr.

However, for purposes of this tutorial we prefer the following evaluator which traverses the structure of the evaluated term. Notice, the natural decomposition of term via unification and build-in functors +,-,*. To simplify the problem, we omit the division (/) operator.
eval(A+B,CV):-eval(A,AV),eval(B,BV),CV is AV+BV.

eval(A-B,CV):-eval(A,AV),eval(B,BV),CV is AV-BV.

eval(A*B,CV):-eval(A,AV),eval(B,BV),CV is AV*BV.

eval(Num,Num):-number(Num).

Now, we can easily extend the above program to allow "variables" in the evaluated term. These variables are represented by Prolog atoms like a, b or c, so they do not correspond to Prolog variables. Ofcourse, we have to notify the values of the variables to the evaluating program. Thus, the list of pairs variable/value as well as the evaluated expression is passed to the evaluator. To get the value of given variable we utilize the function member that is defined in one of previous lectures.
eval_v(A+B,CV,Vars):-eval_v(A,AV,Vars),eval_v(B,BV,Vars),CV is AV+BV.

eval_v(A-B,CV,Vars):-eval_v(A,AV,Vars),eval_v(B,BV,Vars),CV is AV-BV.

eval_v(A*B,CV,Vars):-eval_v(A,AV,Vars),eval_v(B,BV,Vars),CV is AV*BV.

eval_v(Num,Num,Vars):-number(Num).

eval_v(Var,Value,Vars):-atom(Var),member(Var/Value,Vars).

Try ?-eval_v(2*a+b,Val,[a/1,b/5]) to test the above program.

Compiling
Evaluating arithmetic expressions can be easily extended into generating linear code for some abstract stack machine. We use stack machine with following intructions:
· pop(X)- put element X at top of the stack 

· push(X)- remove element X from the top of the stack 

· plus,minus,times(X,Y,Z)- compute the corresponding value Z from numbers X, Y 

· get_value(X,V)- get value of the variable X from memory 

· set_value(X,V)- set value of the variable X in memory 

· X,Y,Z,V are assumed to be permanent registers of the stack machine. 

Note, that we use the accumulator to collect the code and that the code is actually generated from the end to the beginning.
gen_expr(A+B,InCode,OutCode):-

   gen_expr(B,[pop(X),pop(Y),plus(X,Y,Z),push(Z)|InCode],TempCode),

   gen_expr(A,TempCode,OutCode).

gen_expr(A-B,InCode,OutCode):-

   gen_expr(B,[pop(X),pop(Y),minus(X,Y,Z),push(Z)|InCode],TempCode),

   gen_expr(A,TempCode,OutCode).

gen_expr(A*B,InCode,OutCode):-

   gen_expr(B,[pop(X),pop(Y),times(X,Y,Z),push(Z)|InCode],TempCode),

   gen_expr(A,TempCode,OutCode).

gen_expr(Num,InCode,[push(Num)|InCode]):-number(Num).

gen_expr(Var,InCode,[get_value(Var,Value),push(Value)|InCode]):-atom(Var).

If we can generate the code for evaluating expressions it is easy to add generator for assignment. The compiled program is a list of assignments then.
gen_prog([A=Expr|Rest],InCode,Code):-

   atom(A),

   gen_prog(Rest,InCode,TempCode),

   gen_expr(Expr,[pop(X),set_value(A,X)|TempCode],Code).

gen_prog([],Code,Code).

Now, we write an interpreter of generated machine code. The interpreter uses Stack to evaluate arithmetic expressions and Memory to remember values of variables. The Prolog code of the interpreter follows naturally the sematics of used instructions: pop, push, plus, minus, times, get_value and set_value.
eval_prog([push(X)|Code],Stack,Memory):-

   eval_prog(Code,[X|Stack],Memory).

eval_prog([pop(X)|Code],[X|Stack],Memory):-

   eval_prog(Code,Stack,Memory).

eval_prog([plus(X,Y,Z)|Code],Stack,Memory):-

   Z is X+Y,

   eval_prog(Code,Stack,Memory).

eval_prog([minus(X,Y,Z)|Code],Stack,Memory):-

   Z is X-Y,

   eval_prog(Code,Stack,Memory).

eval_prog([times(X,Y,Z)|Code],Stack,Memory):-

   Z is X*Y,

   eval_prog(Code,Stack,Memory).

eval_prog([get_value(X,Value)|Code],Stack,Memory):-

   member(X/Value,Memory),

   eval_prog(Code,Stack,Memory).

eval_prog([set_value(X,Value)|Code],Stack,Memory):-

   set_value(X,Value,Memory,NewMemory)

   eval_prog(Code,Stack,NewMemory).

eval_prog([],Stack,Memory):-

   print_memory(Memory).

The setting value of the variable is not so straightforward as getting the value of the variable using member. If the variable is in the memory, its value has to be changed, otherwise a new pair variable/value is added to the memory.
set_value(X,Value,[X/_|T],[X/Value|T]).

set_value(X,Value,[Y/V|T],[Y/V|NewT):-

   X\=Y,set_value(X,Value,T,NewT).

set_value(X,Value,[],[X/Value]).

Finally, when the interpreter eval_prog finds the end of the code which is indicated by empty list, it prints the contents of the memory, i.e., the values of all variables which are used in the program.
print_memory([X/Value|T]):-

   write(X=Value),nl,print_memory(T).

print_memory([]):-nl.

To encapsulate the compiler/code generator and interpreter/code evaluator, we introduce the following clause.
run(Prog):-

   gen_prog(Prog,[],Code),

   eval_prog(Code,[],[]).

You can try ?-run([a=5,b=a+2,a=3*a+b]) to test the program. But, what if one uses the variable vith undefined value, e.g., ?-run([a=b+2])? The program fails. Improve the program in such a way that it will print a message notifying undefined variables during interpretaion or, better, it will detect the undefined variables during compilation.

Optimizing
Look at the code generated by gen_prog. Is it possible to optimize the code in some way? Ofcourse, it is possible. Here is an example of such trivial optimizer which removes all successive pairs push-pop and unifies their arguments. It is clear that if one element is pushed to a stack and the other element is poped from the same stack immediately then both elements are same (therefore unification).
optimize([push(X),pop(Y)|T],OptCode):-

   X=Y,

   optimize(T,OptCode).

optimize([H|T],[H|OptCode]):-

   optimize(T,OptCode).

optimize([],[]).

Now, we insert optimizer between generator and executor to get optimized program runner.
opt_run(Prog):-

   gen_prog(Prog,[],Code),

   optimize(Code,OptCode),

   eval_prog(OptCode,[],[]).



Do you like the above presented application? If so, you can further extend it to develop a complete compiler and executor of chosen programming language. For example, think about incorporating if-then-else construct into the above language. 



PROLOG is perfect programming language for handling symbolic data. 

	Boolean Expressions 





This chapter extends working with expressions into a different dimension. First, we write a program to evaluate boolean expression which is similar to evaluation of arithmetic expressions. In the second part, we will work with expression in a symbolic manner and we write a program for transformation of a boolean expression into a conjunctive normal form.

Operator definitons
Before we start to work with boolean (logic) expressions, we define some operators which simplify entry of such expressions.
:-op(720,fy,non).

:-op(730,yfx,and).

S:-op(740,yfx,or).

Now, we can write (non a and b) instead of more cumbersome and(non(a),b)
We can also define meta-operations which can be fully transformed into classic operations and, or, non.
:-op(710,yfx,implies).

:-op(710,yfx,equiv).

:-op(740,yfx,xor).

Evaluating
During evaluation of arithmetic expression we exploit the build-in evaluator is which computes value of numeric expression. Now, we have to define procedures for evaluating and, or, not operations in Prolog.
and_d(false,true,false).

and_d(false,false,false).

and_d(true,false,false).

and_d(true,true,true).

or_d(false,true,true).

or_d(false,false,false).

or_d(true,false,true).

or_d(true,true,true).

non_d(true,false).

non_d(false,true).

We should also indicate which values can be used in expressions.
logic_const(true).

logic_const(false).

Now, it is easy to write an evaluator for boolean expressions.
eval_b(X,X):-logic_const(X).

eval_b(X and Y,R):-eval_b(X,XV),eval_b(Y,YV),and_d(XV,YV,R).

eval_b(X or Y,R):-eval_b(X,XV),eval_b(Y,YV),or_d(XV,YV,R).

eval_b(non X,R):-eval_b(X,XV),non_d(XV,R).

Evaluation of meta-operations is transformed into evaluation of classic operations in an obvious way.
eval_b(X implies Y,R):-eval_b(Y or non X, R).

eval_b(X equiv Y,R):-eval_b(X implies Y and Y implies X, R).

eval_b(X xor Y,R):-eval_b((X and non Y) or (Y and non X), R).

Normalizing
In this section, we will write a Prolog program for transformation of boolean expressions into conjunctive normal form. Conjunctive normal form is an expression in the following form:
(a or non b) and c and (b or d or non c).
Note, that we work with arbitrary atoms in boolean expressions now and these atoms are not interpreted, i.e., we do not know their value (true or false).
First, we remove meta-expressions, i.e., implies, equiv & xor which are substituted by or, and & non.
ex2basic(X implies Y, R):-ex2basic(Y or non X,R).

ex2basic(X equiv Y,R):-ex2basic(X implies Y and Y implies X, R).

ex2basic(X xor Y,R):-ex2basic((X and non Y) or (Y and non X), R).

ex2basic(X or Y, XB or YB):-ex2basic(X,XB),ex2basic(Y,YB).

ex2basic(X and Y, XB and YB):-ex2basic(X,XB),ex2basic(Y,YB).

ex2basic(non X, non XB):-ex2basic(X,XB).

ex2basic(X,X):-atom(X).

Second, we move negation non to atomic formulas.
non2basic(non (X and Y),XN or YN):-non2basic(non X,XN),non2basic(non Y,YN).

non2basic(non (X or Y),XN and YN):-non2basic(non X,XN),non2basic(non Y,YN).

non2basic(non non X, XB):-non2basic(X,XB)

non2basic(non X,non X):-atom(X).

non2basic(X and Y,XB and YB):-non2basic(X,XB),non2basic(Y,YB).

non2basic(X or Y,XB or YB):-non2basic(X,XB),non2basic(Y,YB).

non2basic(X,X):-atom(X).

Finally, we can construct a conjunctive normal form.
ex2conj(X and Y,XC and YC):-ex2conj(X,XC),ex2conj(Y,YC).

ex2conj(X or Y, R):-ex2conj(X,XC),ex2conj(Y,YC),join_disj(XC,YC,R).

ex2conj(non X,non X).

ex2conj(X,X):-atom(X).

join_disj(X and Y,Z,XZ and YZ):-join_disj(X,Z,XZ),join_disj(Y,Z,YZ).

join_disj(X,Y and Z,XY and XZ):-X\=(_ and _),join_disj(X,Y,XY),join_disj(X,Z,XZ).

join_disj(X,Y,X or Y):-X\=(_ and _),Y\=(_ and _).

Now, we join above three procedures into compact form which transforms arbitrary expression into its conjuctive normal form. We call this process normalization.
normalize(Ex,Norm):-

   ex2basic(Ex,Bas),

   non2basic(Bas,NonBas),

   ex2conj(NonBas,Norm).

Try to optimize the resulting conjuctive normal form by removing disjunctions containing literal and its negation (e.g., a or non a, which is known to be true). Write similar procedure(s) for transformation into disjunctive normal form.
	Graphs in Prolog 





Graph is another data structure that is widely used in current algorithms. In this lecture we will describe a representation of graphs in Prolog and we will develop some programs for typical graph operations (coloring, search).

Representation
Graph is usually defined as pair (V,E), where V is a set of vertices and E is a set of edges. There are many possible representations of graphs in Prolog, we will show two of them.
Representation A keeps vertices and edges in two different lists (sets):
g([Vertex, ...],[e(Vertex1,Vertex2,Value), ...])

Note, that this representation is appropriate for directed graphs as well as for non-directed graphs. In case of non-directed graphs, one can add each non-directed edge e(V1,V2,H) as two directed edges e(V1,V2,H), e(V2,V1,H) or, better, it is possible to adjust the access procedure edge (defined below).
Representation B is based on idea of neighbourhood and graph is represented as a list of vertices and its neighbourhood.
[Vertex-[Vertex2-Value, ...], ...]

In this case, the representation of non-directed graphs contains each edge two times.
Here is the procedure for access to edges in Representation A.
edge(g(Es,Vs),V1,V2,Value):-

   member(e(V1,V2,Value),Vs).

If the graph is non-directed, the procedure edge can be adjusted in a following way:
edge(g(Es,Vs),V1,V2,Value):-

   member(e(V1,V2,Value),Vs) ; member(e(V2,V1,Value),Vs).

Here is the procedure edge for Representation B.
edge(Graph,V1,V2,Value):-

   member(V1-NB,Graph),

   member(V2-Value,NB).

Now, it is possible to define procedure for finding neighbourhood of the vertex using procedure edge.
neighbourhood(Graph,V,NB):-

   setof(V1-E,edge(Graph,V,V1,E),NB).

In case of Representation B it is better (more efficient) to define neighbourhood directly.
neighbourhood(Graph,V,NB):-

   member(V-NB,Graph).

Note, that some graphs does not use values of edges while other graphs assign values also to vertices. In these cases, the above procedures have to be rewritten accordingly.

Coloring
The goal of graph coloring is to add a color (from limited palette of colors) to each vertex in such a way that the adjacent vertices (via edge) have assigned different colors. Even if the graph coloring seems to be a theoretical-only problem, the algorithms for graph coloring are widelly used in practical applications (constraint satisfaction).
In this lecture we will present three algorithms for graph coloring. We start with naive algorithm that implements generate and test method in rough form. Then we improve the algorithm by joining the generate and test phases into one procedure. Finally, we implement a more sophisticated method called forward checking.
The following program uses generate and test method to color vertices of graph. First, the color is assigned to each vertex and then the program tests the validity of coloring.
% coloring1(+Graph,+Colors,-Coloring)

coloring1(g(Vs,Es),Colors,Coloring):-

   gener(Vs,Colors,Coloring),

   test(Es,Coloring).

% gener(+Vertices,+Colors,-Coloring)

gener([],_,[]).

gener([V|Vs],Colors,[V-C|T]):-

   member(C,Colors), % non-deterministic generator of colors

   gener(Vs,Colors,T).

% test(+Edges,+Coloring)

test([],_).

test([e(V1,V2)|Es],Coloring):-

   member(V1-C1,Coloring), % find color of vertex V1

   member(V2-C2,Coloring), % find color of vertex V2

   C1\=C2,                 % test the difference of colors

   test(Es,Coloring).

The above program is not very efficient because it generates many wrong colorings which are rejected in the testing phase. In addition, the generator leaves out the conflicting vertices and it generates other colorings independently of the conflict.
It is clear that we can test validity of coloring during generation of colors. Following program joins generation and testing into one procedure. Note, that we use accumulator to save the partial coloring.
% coloring2(+Graph,+Colors,-Coloring)

coloring2(g(Vs,Es),Colors,Coloring):-

   gat(Vs,Es,Colors,[],Coloring).   % generate and test

% gat(Vertices,Edges,Colors,ColoredVertices,FinalColoring)

gat([],_,_,Coloring,Coloring).

gat([V|Vs],Es,Cs,Acc,Coloring):-

   member(C,Cs),          % generate color for vertex V

   test2(Es,V,C,Acc),     % test the validity of current coloring

   gat(Vs,Es,Cs,[V-C|Acc],Coloring).

% test2(+Edges,+Vertex,+Color,+CurrentColoring)

test2([],_,_,_).

test2([e(V1,V2)|Es],V,C,CColoring):-

   (V=V1 -> (member(V2-C2,CColoring) -> C\=C2 ; true)

    ;(V=V2 -> (member(V1-C1,CColoring) -> C\=C1 ; true)

     ;true)),

   test2(Es,V,C,CColoring).

The above program uses backtracking to find another valid coloring, but it is not able to detect conflict before the conflict really occurs, i.e., after assigning the color to the second vertex of the conflicting edge.
It is possible to improve behaviour of the algorithm by forward checking of conflicts. First, we assign the set of all possible colors to each vertex (prep). Then, we choose one vertex and its color (from the set of possible colors assigned to this vertex) and we remove this color from all adjacent vertices (fc), i.e., we remove (some) future conflicts. Therefore, we know that the assigned color is not in conflict with already colored vertices.
Note, that as the forward checking adds some addtional overhead to the algorithm, it is possible that the classical backtracking could be more efficient in some cases. Also, the efficiency of forward checking algorithm depends on the strategy of choosing variables and colors for assignment.
% coloring3(+Graph,+Colors,-Coloring)

coloring3(g(Vs,Es),Colors,Coloring):-

   prep(Vs,Colors,ColoredVs),

   gtb(ColoredVs,Es,[],Coloring).

% prep(+Vertices,+Colors,+SuperColoring)

prep([],_,[]).

prep([V|Vs],Colors,[V-Colors|CVs]):-

   prep(Vs,Colors,CVs).

% gtb(+SuperColoring,+Edges,+PartialColoring,-Coloring)

gtb([],_,Coloring,Coloring).

gtb([V-Cs|Vs],Es,Acc,Coloring):-

   member(C,Cs),                  % select only one color

   fc(Es,V,C,Vs,ConstrainedVs),   % forward checking

   gtb(ConstrainedVs,Es,[V-C|Acc],Coloring).

% fc(+Edges,+Vertex,+VertexColor,+InputSuperColoring,-OutputSuperColoring)

fc([],_,_,Vs,Vs).

fc([e(V1,V2)|Es],V,C,Vs,ConstrVs):-

   (V=V1 -> constr(Vs,V2,C,NewVs)

    ;(V=V2 -> constr(Vs,V1,C,NewVs)

      ;NewVs=Vs)),

   fc(Es,V,C,NewVs,ConstrVs).

% constr(+InputSuperColoring,+Vertex,-VertexForbiddenColor,+OutputSuperColoring)

constr([V-Cs|Vs],V,C,[V-NewCs|Vs]):-

   delete(Cs,C,NewCs),NewCs\=[].

constr([V1-Cs|Vs],V,C,[V1-Cs|NewVs]):-

   V\=V1,

   constr(Vs,V,C,NewVs).

constr([],_,_,[]).

delete([],_,[]).

delete([X|T],X,T).

delete([Y|T],X,[Y|NewT]):-

   X\=Y,

   delete(T,X,NewT).

Note, that delete does not fail if the element is not present in the list.

Search
Another popular group of algorithms regarding graphs is search. In this lecture we will present two algorithms: simple search that finds path between two vertices and Dijkstra's algorithm which finds minimal distance/path from one vertex to all vertices.
The following program finds a path from one vertex to another vertex. The same program can be used to find path in both directed and non-directed graphs depending on the definiton of procedure edge. Note, that we use accumulator containing part of path to prevent cycles.
% path(+Graph,+Start,+Stop,-Path)

path(Graph,Start,Stop,Path):-

   path1(Graph,Start,Stop,[Start],Path).

path1(Graph,Stop,Stop,Path,Path).

path1(Graph,Start,Stop,CurrPath,Path):-

   Start\=Stop,

   edge(Graph,Start,Next),

   non_member(Next,CurrPath),

   path1(Graph,Next,Stop,[Next|CurrPath],Path).

non_member(_,[]).

non_member(X,[Y|T]):-

   X\=Y,

   non_member(X,T).

Dijkstra's algorithm is a well known algorithm for finding minimal path in graphs with (non-negative) edges. Here is its implementation in Prolog which finds minimal distance to all vertices from given vertex.
% min_dist(+Graph,+Start,-MinDist)

min_dist(Graph,Start,MinDist):-

   dijkstra(Graph,[],[Start-0],MinDist).

% dijkstra(+Graph,+ClosedVertices,+OpenVertices,-Distances)

dijkstra(_,MinDist,[],MinDist).

dijkstra(Graph,Closed,Open,MinDist):-

   choose_v(Open,V-D,RestOpen),

   neighbourhood(Graph,V,NB),  % NB is a list of adjacent vertices+distance to V

   diff(NB,Closed,NewNB),

   merge(NewNB,RestOpen,D,NewOpen),

   dijkstra(Graph,[V-D|Closed],NewOpen,MinDist).

% choose_v(+OpenVertices,-VertexToExpand,-RestOpenVertices)

choose_v([H|T],MinV,Rest):-

   choose_minv(T,H,MinV,Rest).

choose_minv([],MinV,MinV,[]).

choose_minv([H|T],M,MinV,[H2|Rest]):-

   H=V1-D1, M=V-D,

   (D1<D -> NextM=H,H2=M

          ; NextM=M,H2=H),

   choose_minv(T,NextM,MinV,Rest).

% diff(+ListOfVertices,+Closed,-ListOfNonClosedVertices)

diff([],_,[]).

diff([H|T],Closed,L):-

   H=V-D,

   (member(V-_,Closed) -> L=NewT ; L=[H|NewT]),

   diff(T,Closed,NewT).

% merge(+ListOfVertices,+OldOpenVertices,-AllOpenVertices)

merge([],L,_,L).

merge([V1-D1|T],Open,D,NewOpen):-

   (remove(Open,V1-D2,RestOpen)

      -> VD is min(D2,D+D1)

       ; RestOpen=Open,VD is D+D1),

   NewOpen=[V1-VD|SubOpen],

   merge(T,RestOpen,D,SubOpen).

remove([H|T],H,T).

remove([H|T],X,[H|NT]):-

   H\=X,

   remove(T,X,NT).

Compare the procedure remove with the procedure delete (coloring part). Do you see the difference?
Extend the above program in such a way that it also finds the minimal path (not only the minimal distance) to all vertices.
	How does it work? 





In this lesson I try to collect some features of Prolog language which I have found to be difficult to understand by students accustomed to procedural style of programming.


is 

"is" is a build-in arithmetic evaluator in Prolog. "X is E" first computes the arithmetic expression E and then unifies the result with X. E can contain variables but these variables has to be bound to numbers, e.g., "X=5, Y is 2*X" is correct Prolog goal. Note, that we can use unification to bind variables but do not interchange unification and evaluation. Also, do not use "is" in the same way as assignment in procedural languages! Following examples shows wrong usage of "is": 

?-X is Y+2.              % arithmetic expression cannot be evaluated because Y is free

?-H=5,T=[], L is [H|T].  % expression [H|T] is not arithmetic

?-X is 2, X is 3.        % fails because it is impossible to change the value of X from 2 to 3



unification 

Unification is used to bind variables as well as to compare data structures in both directions. Unification does not evaluate expressions. 

?-X=f(Y).         % returns X/f(Y)

?-f(g(Y))=f(X).   % returns X/g(Y)

?-X=1+2.          % returns X/1+2

?-3=1+2.          % fails (3 is syntactically different from 1+2)



backtracking 

Backtracking is a powerful feature of Prolog that simplifies development of many programs. It enables the program to use other alternative if the previous alternative fails. Thus, programming of generate and test algorithms is natural in Prolog. Also, it is usually possible to find one solution as well as all solutions using the same program. 

  



declarative character 

Declarative character of many Prolog programs enables one to use the same procedure in different ways. Note, that it is not distinguished whether the argument of the procedure is input or output in Prolog. Thus, it is possible to use one argument as input in one call and use the same argument as output in other call. See following example: 

member(X,[X|T]).

member(X,[_|T]):-member(X,T).

?-member(1,[1,2,3]).  % usage as a test

?-member(X,[1,2,3]).  % usage as a member generator (returns successively X=1, X=2, X=3)

?-member(1,L).        % usage as a list generator (returns L=[1|_], L=[_,1|_], L=[_,_,1|_] etc.)

?-member(X,L).        % generator of general lists containing X (returns L=[X|_], L=[_,X|_], L=[_,_,X|_] etc.)



operators 

Operators simplify entry of Prolog programs. They are only used to define syntactic conventions of program and data entry. The definition of operators helps Prolog to understand expression like 1+2+3*4 which is translated into notation +(+(1,2),*(3,4)). 

  



cut 

Cut is a feature of Prolog (not logic programming) that is used to cut alternative branches of computation and, thus, these branches are not explored by backtracking. Cut can improve efficiency of Prolog programs, however, it also changes the clear operational behaviour of programs. Use "cut" carefully as the programs containing cut are usually harder to read. The following example explains the behaviour of cut. 

?-member(Y,[[1,2],[3,4]]),member(X,Y).     % returns X=1,X=2,X=3,X=4 successively

?-member(Y,[[1,2],[3,4]]),member(X,Y),!.   % returns X=1 only

?-member(Y,[[1,2],[3,4]]),!,member(X,Y).   % returns X=1, X=2 successively

?-!,member(Y,[[1,2],[3,4]]),member(X,Y).   % returns X=1,X=2,X=3,X=4 successively



negation 

Because of complexity reasons, Prolog does not contain full logic negation. Instead of it, Prolog uses negation as failure which is based on Closed World Assumption. The operational behaviour of this type of negation can be expressed by the following program: 

not P:-P,!,fail.

not P.

Negation in Prolog can be safely used only as a test. Assure that all variables in negated goal are bound, otherwise you can get "strange" results as following examples show: 

p(a).

p(b).

q(c).

?-not p(X), q(X).   % fails

?-q(X), not p(X).   % succeeds with X=c



If you do not understand how does something work, then try some simple examples.
	Meta-Programming 





Meta-programming is a programming technique that enables manipulation with program structures. Because Prolog uses the same data structures to represent programs as well as data, Prolog is suitable for writing meta-programs.


In the section "Prolog Data Structures", we discussed terms, basic Prolog data structure, and some predicates (arg,functor,=..) which work with terms. Now we extend this set of predicates by predicate "call" that is used to call arbitrary Prolog goal. In fact, one can build a Prolog goal using =.. and then use call to execute this goal. 

In the following code we use the features of =.. and call to design a procedure which can apply (map) given function to each element of list. The name of this function is an input parameter of the procedure.
map(FunctionName,[H|T],[NH|NT]):-

   Function=..[FunctionName,H,NH],

   call(Function),

   map(FunctionName,T,NT).

map(_,[],[]).

Now, one can call
?-map(neg,[1,2,3],L).          % result is L=[-1,-2,-3]

where neg is defined in a following way:
neg(A,B):-B is -A.

to get the procedure that negates elements of list.
The same procedure map can be used to get other list processors by adding code of functions, e.g.,
inc(A,B):-B is A+1.

dec(A,B):-B is A-1.

and goal:
?-map(inc,[1,2,3],X),map(dec,X,Y).      % result is X=[2,3,4], Y=[1,2,3]

Compare the above approach with generalized list processor.
	Meta-Interpreters 





Because it is possible to directly access program code in Prolog, it is easy to write interpreter of Prolog in Prolog. Such interpreter is called a meta-interpreter. Meta-interpreters are usually used to add some extra features to Prolog, e.g., to change build-in negation as failure to constructive negation.
The simplest Prolog meta-interpreter is a following program:
solve(Goal):-call(Goal).

However, there is not advantage of using such meta-intepreter as it immediately calls Prolog interpreter. Much more popular is "vanilla" meta-interpreter that uses Prolog's build-in unification but enables access to search engine which can be easily modified (e.g., it is possible to change the order of goals' execution)
solve(true).

solve((A,B)):-

   solve(A),solve(B).

solve(A):-

   clause(A,B),solve(B).

Note, that vanilla meta-interpreter uses build-in predicate clause(H,B) which finds a clause in Prolog program with head that unifies with H and body B (if there is no body, then Body=true).
The modified vanilla meta-interpreter can be used to compute "proof" of the computation:
solve(true,fact).

solve((A,B),(ProofA,ProofB)):-

   solve(A,ProofA),solve(B,ProofB).

solve(A,A-ProofB):-

   clause(A,B),solve(B,ProofB).

It is also possible to write a meta-interpreter that uses list of goals instead of traditional conjunction of goals. In some cases, this could be more natural as one does not need to traverse the structure of goal each time a primitive goal is being found.
solve([]).

solve([A|T]):-

   clause(A,B),

   add_to_list(B,T,NT),

   solve(NT).

We modify the above meta-interpreter to get a meta-interpreter of guarded Prolog. Each clause in guarded Prolog contains a test (Prolog goal) called guard which has to be satisfied before using the clause. Because traditional Prolog has no guards we can write meta-interpreter that extends behaviour of Prolog to handle guards.
solve([]).

solve([A|T]):-

   clause(A,B),

   distribute(B,GoalB,Guard),

   call(Guard),

   append(GoalB,T,NT),

   solve(NT).

Let the following clause be the clause with guards, where guards is indicated by g:
a(X):-g(X>0),b(X).

Then , the procedure distribute that finds guards and regual goals in the body of clause can be programmed in a following way (using double accumulator):
distribute(B,ListGoal,Guard):-

   distr(B,[],ListGoal,true,Guard).

distr((A,B),OldList,NewList,OldG,NewG):-          % conjunction

   distr(B,OldList,SubList,OldG,SubG),

   distr(A,SubList,NewList,SubG,NewG).

distr(g(G),GoalList,GoalList,OldG,(G,OldG)).      % guard

distr(G,GoalList,[G|GoalList],Guard,Guard):-      % regular primitive goal

   G\=(_,_),G\=g(_).

Because it is possible to directly access program code in Prolog, it is easy to write interpreter of Prolog in Prolog. Such interpreter is called a meta-interpreter. Meta-interpreters are usually used to add some extra features to Prolog, e.g., to change build-in negation as failure to constructive negation.
Exercises 

1. A simple Prolog session 

Create an elementary Prolog knowledge base, describing following relations: 

1. John, Fred and Harry are men, Mary, Julie, Susan and Anne are women. 

2. John has blonde hair while Fred and Harry have dark hair. 

3. Julie and Susan are blonde, Mary and Anne are brunette. 

4. Rich is each person who owns the gold - Fred and Julie in our example. 

5. Male like only female and vice versa. Moreover, John and Harry like rich persons, John likes blonde and Fred likes brunette. 

6. Both Mary and Julie like dark hair persons, Julie likes rich persons at the same time. 

7. Anne owns a house and John owns a car. 

and ask it: 

a. Who is male ? 

b. Who does John like ? 

c. Who does Mary like ? 

d. Does Julie like anyone ? 

e. Is there a couple who like each other ? 

2. Graph search in Prolog. 

Let us have following situation:

Fig. 1 Processed graph

 

1. Determine if a path exists in a directed, acyclic graph (Fig. 1). 

2. Determine paths and path lengths in a directed, acyclic graph (Fig. 1). 

3. Print paths to the screen in a directed, acyclic graph (Fig. 1). 

4. Print all paths to the screen in a directed, acyclic graph (Fig. 1). 

5. Paths and all paths in a directed, possibly cyclic, graph (additional edge ea in (Fig. 1). 

3. Cryptarithmatic

Question: 

1. What is your opinion about this program ? 

2. How it works and what is the results? 

3. What the program does to minimalize the search space of the problem ? 
% A sample crytarithm problem 

% Shows the use of a list of variables to keep track of whether

% they have been assumed to have values or not yet.

:-op(100, xfx, not_in).

_ not_in [].

D not_in [X | Rest]:- var(X),! , %added to reduce trail


D not_in Rest.

D not_in [X | Rest]:- nonvar(X), D\==X,!, %added to reduce trail


 D not_in Rest.

digit(0). digit(1). digit(2). digit(3). digit(4). digit(5). digit(6).

digit(7). digit(8). digit(9).

carry(0). carry(1).

go:-send_more_money([S,E,N,D,M,O,R,Y]),result([S,E,N,D,M,O,R,Y]).

result([S,E,N,D,M,O,R,Y]):-


nl,write('  S E N D'),tab(5),  out([' ', S, E, N, D]),


nl,write('  M O R E'), tab(5), out([' ', M, O, R, E]),


nl,write('---------'), tab(5), write(--------------),


nl,write('M O N E Y'), tab(5), out([M, O, N, E, Y]),


nl.

out([]).

out([C|Rest]):-write(C), write(' '), out(Rest).

send_more_money(DigitList):-


DigitList=[S,E,N,D,M,O,R,Y],


M=1,

  % because M is carry into most significant


digit(S), S\=0,  % because S is most significant digit


column(1,0,0,C2,M,0,DigitList),  %Cn are the carries


column(2,S,M,C3,O,C2,DigitList),


column(3,E,O,C4,N,C3,DigitList),


column(4,N,R,C4,E,C5,DigitList),


column(5,D,E, 0,Y,C5,DigitList).

column(Col,DigitInLine1, DigitInLine2, Carry, DigitInLine3, NewCarry, DigitList):-


carry(Carry), %try 0 and 1 in turn


try_digit(Col,1, DigitInLine1,DigitList),


try_digit(Col,2, DigitInLine2,DigitList),


Sum is DigitInLine1+DigitInLine2+Carry,


NewCarry is Sum // 10,


X3 is Sum - (10*NewCarry),


(var(DigitInLine3), X3 not_in DigitList


;  nonvar(DigitInLine3)


),


DigitInLine3=X3.

try_digit(Column,Line,Digit,DigitList):- nonvar(Digit),!. 




% cuts added to reduce trail




% Column and Line are for tracing

try_digit(Column,Line,Digit,DigitList):-


var(Digit),!, digit(X1), X1 not_in DigitList, Digit=X1.

:- write(''), nl.

:- write(''), nl.

:- write('The puzzle SEND+MORE=MONEY has been loaded'), nl.

:- write('To solve it try go. ...'), nl.

4. Region Coloring

A famous problem in mathematics concerns coloring adjacent planar regions. Like cartographic maps, it is required that, whatever colors are actually used, no two adjacent regions may not have the same color. Two regions are considered adjacent provided they share some boundary line segment. Consider the following map. 

[image: image14.png]



We have given numerical names to the regions. To represent which regions are adjacent, consider also the following graph. 

[image: image15.png]



Here we have erased the original boundaries and have instead drawn an arc between the names of two regions, provided they were adjacent in the original drawing. In fact, the adjacency graph will convey all of the original adjacency information. A Prolog representation for the adjacency information could be represented by the following unit clauses, or facts.

PAGE  
29

